code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_19:街角（v1.1）
;-------------------------------------------------
@QST_TITLE_QUEST_19
RESULTS '= "街角"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_19
RESULTS '= "周期,街,"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_19
RESULTS:0 '= "主人公の広大な領地からほど近い場所に、とある小さな町がある"
RESULTS:1 '= "森と丘陵に囲まれた穏やかなこの町は、"
RESULTS:2 '= "四季折々の美しい木々と花々に彩られ、穏やかな日常が営まれている"
RESULTS:3 '= ""
RESULTS:4 '= "道沿いには店が軒を連ね、近隣の農村からの農産物や手工芸品が取引されている"
RESULTS:5 '= "町の北には地元の人々に長く愛されてきた小さな公園がある"
RESULTS:6 '= ""
RESULTS:7 '= "特別な観光スポットや名所はないが、穏やかな散歩やリフレッシュには最適だろう"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_19
RESULTS:0 '= @"難易度：- (戦闘無しの町シナリオ)"
RESULTS:1 '= "報酬：-"
RESULTS:2 '= "推奨パーティ人数:1人以上"
RESULTS:3 '= @"@C:{カラー_ライトブルー}@★自由に帰還可能@/C@"
RESULTS:4 '= ""
RESULTS:5 '= @"@C:{カラー_選択不可}@「大したものはないが、いい町だ」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「のんびりと過ごす分にはいいところだよ」@/C@"
RESULTS:7 '= @"@C:{カラー_選択不可}@「いつまでもこんな日が続けばいいんだが……」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_19(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_0_ゴブリンの洞窟の攻略が必要
SIF !QST_SV_QUEST_FLAG:0
    RETURN 0
RETURN 1

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_19(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_19_街角"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "街角"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"                 ;入口
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,12,11,]"              ;道１/普通
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,20,21,23,22,]"     ;道２/賑わい
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,30,31,]"           ;道３/静かな
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,40,41,42,43,44,]"  ;店の前
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,50,51,52,53,54,59,]";怪しい店
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,60,61,62,]"        ;公園
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,71,72,73,]"        ;公園奥
;-------------------------------------------------
;！各マップメモ
;* ◆0:入口
;* ◆1:道１/普通
;[11]占い師
;[12]忙しそうな男性
;* ◆2:道２/賑わい
;[20]サンドイッチ屋
;[21]噴水
;[22]噴水前広場
;[23]ベンチ
;* ◆3:道３/静かな
;[30]施設-路地裏/900
;[31]吟遊詩人
;* ◆4:店の前
;[40]店を見て時間を潰す
;[41]ベンチ
;[42]小さな店
;[43]雑貨屋（冒険者っぽいもの。武器も売ってる）
;[44]服屋
;* ◆5:怪しい店
;[50]商品を購入する
;[51]宝石の飾られた棚
;[52]商品の並んだ壁
;[53]悩む女性
;[54]商品を見る男性
;[59]店に入った時のテキスト
;* ◆6:公園
;[60]施設-ベンチ/22
;[61]施設-広場/21
;[62]施設-トイレ/9
;* ◆7:公園奥
;[71]施設-茂み/12
;[72]施設-ベンチ/22
;[73]施設-茂み(木々)/12

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_19
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_19

;-------------------------------------------------
;クエストの探索状態保持設定
;-------------------------------------------------
@QST_QUEST_19_SAVE_EXPLORED
RETURN 1

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_19_ABORT
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;クエストの導入文
;------------
@QST_INTRO_QUEST_19
CALL PRINT_TITLE("街角", "クエスト")
PRINTFORML 近くの町へと向かうことにした……
PRINTFORML …………
PRINTFORML ……
PRINTW
SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTFORML 町の周囲には桜の木々が立ち並び、鳥たちのさえずりが聞こえる
        PRINTFORML 赤い花や白い花が咲き乱れ、風にその花弁を揺らす
        PRINTL
        PRINTFORML 辺りは暖かな光につつまれ、穏やかな空気が漂っている……
    CASE 1 ;夏
        PRINTFORML 夏の陽光が町を照らす……
        PRINTL
        PRINTFORML 町は緑に包まれ、木々の葉が風に揺れる
        PRINTFORML 暑さをしのぐため、町の人々は涼しい木陰でくつろいでいるようだ
    CASE 2 ;秋
        PRINTFORML ……柔らかな日差しが町を包んでいる
        PRINTL
        PRINTFORML 町の周囲は紅葉に染まり、木々が色鮮やかに輝いている
        PRINTFORML 秋風が心地よく吹き抜ける町は、紅葉が美しく街並みを彩っていた
    CASE 3 ;冬
        PRINTFORML 雪景色に包まれた冬の町はいつもより静かだ……
        PRINTFORML かんとした冷たさが肌を刺し、吐く息は白く染まる
        PRINTL
        PRINTFORML 寒さをしのぐため、町の人々は厚着をして雪の中を歩いていた
        PRINTFORML 木々には白い雪が積もり、暖かな灯りが家々から漏れ出している……
ENDSELECT
PRINTW
PRINTFORML 町には素朴な家々が立ち並び、町の中心部にはサンドウィッチの屋台や雑貨屋などが並んでいる
PRINTFORML 少し小高い丘の上にある時計塔は、その鐘の音と共に眠りについた人々を優しく起こしている事だろう
PRINTW
PRINTFORML ……
PRINTFORMW ……さて、どうしようか……

;ゲーム内時刻を8:00にセット
HO_TIME = HO_CLOCK(800)
FLAG:クエスト_中断を許可 = 1
;居住区で実行可能なコマンドの一部を実行可能にするフラグ
FLAG:クエスト_交流コマンド許可 = 1
;0時にクエスト強制終了
FLAG:QST_終了時刻 = HO_CLOCK(2400)
;BGMセット
CALL FL_PLAYBGM("Nostalgic_0.ogg")

;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;エリア入場時にログ出力する情景描写
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_0
; #region EV0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_19_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 0 ;入口
        CALL HO_PRINT_LOG("石畳に覆われた道路に、色とりどりの花々が咲き誇る", 1)
    CASE 1 ;道１/普通
        SIF HO_SEASON() == HO_季節_春
            CALL HO_PRINT_LOG("道端の花々が色とりどりに咲き誇り、麗らかな風が心地よく吹き抜ける", 1)
        SIF HO_SEASON() == HO_季節_夏
            CALL HO_PRINT_LOG("夏の陽射しに照らされた石畳に、緑豊かな木々が涼しげな影を落としている", 1)
        SIF HO_SEASON() == HO_季節_秋
            CALL HO_PRINT_LOG("秋風が吹き抜ける石畳には、落ち葉が敷き詰められている", 1)
        SIF HO_SEASON() == HO_季節_冬
            CALL HO_PRINT_LOG("町は白い雪に覆われ、澄んだ空気が冷たく鼻をくすぐる。", 1)
    CASE 2 ;道２/賑わい
        SIF HO_SEASON() == HO_季節_春
            CALL HO_PRINT_LOG("桜の木々が華やかな花を咲かせている", 1)
        SIF HO_SEASON() == HO_季節_夏
            CALL HO_PRINT_LOG("賑やかな街角では、夏の陽射しに照らされた街路樹が青々と葉を茂らせている", 1)
        SIF HO_SEASON() == HO_季節_秋
            CALL HO_PRINT_LOG("街路樹の葉が赤や黄色に色づき、落ち葉が風に舞っている", 1)
        SIF HO_SEASON() == HO_季節_冬
            CALL HO_PRINT_LOG("雪が積もった街路樹が寒々しく立ち並び、冬の訪れを感じさせる", 1)
    CASE 3 ;道３/静かな
        SIF HO_SEASON() == HO_季節_春
            CALL HO_PRINT_LOG("静かな道に穏やかな風が吹く", 1)
        SIF HO_SEASON() == HO_季節_夏
            CALL HO_PRINT_LOG("街道には静寂が広がり、木々の葉がそよぐ", 1)
        SIF HO_SEASON() == HO_季節_秋
            CALL HO_PRINT_LOG("街道には寂しげな風が吹き抜けている", 1)
        SIF HO_SEASON() == HO_季節_冬
            CALL HO_PRINT_LOG("樹木は雪に包まれ、静かに立ち並んでいる", 1)
    CASE 4 ;店の前
        SIF HO_SEASON() != HO_季節_冬
            CALL HO_PRINT_LOG("石畳の路地には、古びた看板が風に揺れ、犬がのんびりと日向ぼっこをしている", 1)
        SIF HO_SEASON() == HO_季節_冬
            CALL HO_PRINT_LOG("古びた看板に雪が積もっている", 1)
    CASE 5 ;怪しい店
        CALL HO_PRINT_LOG("店には様々な不思議なものが並んでいる……", 1)
    CASE 6 ;公園
        IF HO_OCLOCK() < 1900
            SIF HO_SEASON() == HO_季節_春
                CALL HO_PRINT_LOG("公園では子供たちが元気に遊び、ベンチでは老人たちが寛いでいる", 1)
            SIF HO_SEASON() == HO_季節_夏
                CALL HO_PRINT_LOG("夏の公園は活気にあふれ、芝生では子供たちが楽しそうに遊んでいる", 1)
            SIF HO_SEASON() == HO_季節_秋
                CALL HO_PRINT_LOG("秋の公園は活気にあふれ、落ち葉の絨毯が広がっている", 1)
            SIF HO_SEASON() == HO_季節_冬
                CALL HO_PRINT_LOG("雪景色に覆われた公園で、寒さにも負けず子供たちは雪合戦を楽しんでいる", 1)
        ELSE
            ;夜のテキスト
            SIF HO_SEASON() == HO_季節_春
                CALL HO_PRINT_LOG("花々の香りが淡く漂い、夜風が心地よく吹き抜ける", 1)
            SIF HO_SEASON() == HO_季節_夏
                CALL HO_PRINT_LOG("夜の公園には人の気配はなく、星が輝く夜空が美しい", 1)
            SIF HO_SEASON() == HO_季節_秋
                CALL HO_PRINT_LOG("夜の公園には人の気配はなく、落ち葉が静かに舞い落ちる", 1)
            SIF HO_SEASON() == HO_季節_冬
                CALL HO_PRINT_LOG("夜空の下、公園は雪に覆われて静寂に包まれている", 1)
        ENDIF
    CASE 7 ;公園奥
        IF HO_OCLOCK() < 1900
            SIF HO_SEASON() == HO_季節_春
                CALL HO_PRINT_LOG("花と木々の間を小鳥のさえずりが響き渡り、穏やかな風がそよぐ", 1)
            SIF HO_SEASON() == HO_季節_夏
                CALL HO_PRINT_LOG("青々と茂る木々の間を小鳥のさえずりが響き渡り、そよ風が心地よく吹き抜ける", 1)
            SIF HO_SEASON() == HO_季節_秋
                CALL HO_PRINT_LOG("地面は色とりどりの落ち葉に覆われ、静寂に包まれている", 1)
            SIF HO_SEASON() == HO_季節_冬
                CALL HO_PRINT_LOG("木々の枝も雪で覆われている", 1)
        ELSE
            ;夜のテキスト
            SIF HO_SEASON() == HO_季節_春
                CALL HO_PRINT_LOG("夜の公園に人影はなく、鳥のさえずりだけが聞こえる", 1)
            SIF HO_SEASON() == HO_季節_夏
                CALL HO_PRINT_LOG("夜空の下、虫の鳴き声が遠くに響く", 1)
            SIF HO_SEASON() == HO_季節_秋
                CALL HO_PRINT_LOG("地面は色とりどりの落ち葉に覆われ、月明かりのみが静かに照らしている", 1)
            SIF HO_SEASON() == HO_季節_冬
                CALL HO_PRINT_LOG("静寂の中、月明かりに雪が照らされている", 1)
        ENDIF
ENDSELECT
; #endregion

;-------------------------------------------------
;EV11- 占い師の女性
;-------------------------------------------------
; あとでイベント盛りたいけどとりあえず一文のみで
; あなたには、食べ物をたくさん食べたい/はしゃぎたい/誰にも言えない
; という欲望があるのではありませんか？とか……色々適当な事言ってくる
; キャラデータ一部参照してもいいけど基本的にはランダムを楽しむ感じで
; 一応男か女かはちゃんと見る

@QST_ADD_EVENT_QUEST_19_EVENT_11
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "占い師の女性"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_11
SETCOLOR カラー_ライトブルー
PRINTFORMW 「ありふれた悩みとはいえ、恋愛とか、そういうことに悩めるのは幸せなことだと思うよ」
RESETCOLOR

;-------------------------------------------------
;EV12- 忙しそうな男性
;-------------------------------------------------
; 体力増やせるヒントでも書いておこう

@QST_ADD_EVENT_QUEST_19_EVENT_12
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "忙しそうな男性"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_12
SETCOLOR カラー_ライトブルー
PRINTFORMW 「公園の広場で運動でもして、体力をつけるといいよ！」
PRINTFORMW 「全てにおいて、体力は大事だからね！」
RESETCOLOR

;-------------------------------------------------
;EV20 サンドウィッチの屋台
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_20
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "サンドウィッチの屋台"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_20
;準備
#DIM DYNAMIC TARGET_ID
;描写
PRINTFORML 陳列されたサンドウィッチは、色とりどりの具材で彩られている
PRINTFORMW カウンターの奥には、笑顔の店員が立っている。
PRINTL
SETCOLOR カラー_オレンジ
PRINTFORML 「こんにちは！
PRINTDATAW
    DATAFORM %HO_SEASON_STR:HO_SEASON()%限定のサンドウィッチをお試しになりますか？」
    DATAFORM 季節のサンドウィッチをお試しになりますか？」
    DATAFORM %HO_SEASON_STR:HO_SEASON()%だけの味が楽しめます」
    DATAFORM 季節限定の特別な味わいをお楽しみいただけますよ」
    DATAFORM 焼きたてのパンに、新鮮な季節の野菜と、特製のソースが絡み合って絶品です」
ENDDATA
RESETCOLOR
PRINTL
PRINTFORML ----------------------
PRINTFORML メニュー
PRINTFORML ----------------------
PRINTFORML 季節のサンドウィッチ　値段：3000ディナリ
PRINTFORML サンドウィッチ（お持ち帰り）　値段：2500ディナリ
PRINTL
CALL ASK_MULTI_JUDGE("季節のサンドウィッチを購入する", QST_GET_FLAGMAP("季節サンドウィッチ在庫") != "購入済み", "サンドウィッチを１個購入する", QST_GET_FLAGMAP("お持ち帰り在庫") != "購入済み", "やめておく", 1)
;選択肢後共通。ASK_MULTI_JUDGEでは数字が出るので消しておく。
CLEARLINE 1
PRINTL
;以下選択肢別。お金持ってるか確認した後に分岐
SELECTCASE RESULT
    CASE 0
        IF MONEY < 3000
            PRINTFORMW お金が足りないようだ……
            RETURN
        ENDIF
        CALL QST_QUEST_19_EVENT_20_SEASON_SANDWICH
        CALL FL_PLAYSE("Money_0.mp3")
        CALL COLOR_PRINTL(@"季節のサンドウィッチを購入した！", カラー_注意)
        CALL QST_CHANGE_ITEM("MONEY,-3000,")
        CALL QST_SET_FLAGMAP("季節サンドウィッチ在庫", "購入済み")
        PRINTL
        PRINTFORML %QST_LEADER_NAME()%は季節のサンドウィッチを味わった
        PRINTDATAW
            DATAFORM ……これは美味い！
            DATAFORM 感動の味わいだ！
            DATAFORM 夢中で食べてしまった！
            DATAFORM 贅沢なひとときだった……
            DATAFORM 心が満たされる味わいだ……
        ENDDATA
        PRINTL
        FOR LOCAL, 0, QST_PARTYNUM
            TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
            CALL QST_CHANGE_HP(TARGET_ID, 4000)
        NEXT
        WAIT
    CASE 1
        IF MONEY < 2500
            PRINTFORMW お金が足りないようだ……
            RETURN
        ENDIF
        PRINTFORMW サンドウィッチを１個購入した！
        PRINTL
        CALL FL_PLAYSE("Money_0.mp3")
        CALL QST_CHANGE_ITEM(@"MONEY,-2500,{ITEM_サンドウィッチ},1,")
        CALL QST_SET_FLAGMAP("お持ち帰り在庫", "購入済み")
    CASEELSE
        PRINTFORML ……やめておこう
        PRINTL
        PRINTFORMW サンドウィッチ屋の前を離れた……
        RETURN
ENDSELECT

;;;季節のサンドウィッチ関連のテキストを分割した
@QST_QUEST_19_EVENT_20_SEASON_SANDWICH
;開始部分
PRINTFORMW %QST_LEADER_NAME()%は季節のサンドウィッチを購入することにした
PRINTFORMW さて、どれにしようか……
PRINTL ……
;季節ごと解説文
;あとの部分と足しこむのでDATAFORM
SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTDATA
            DATAFORM フレッシュなアボカド、レタス、トマトにバジルペーストを合わせた
            DATAFORM 甘酸っぱいベリーとクリームチーズとふわふわのパンの
            DATAFORM ハーブ風味のパンに新鮮なレタスとベーコンを挟んだ
            DATAFORM 春の陽気なピクニックにぴったりの鶏肉とレタスの
            DATAFORM バゲットに、新鮮なルッコラとトマトとハムを挟んだ
            DATAFORM 粗挽き胡椒とマヨネーズがアクセントのサーモンとレタスの
            DATAFORM ビネガーと塩胡椒で下味をつけたキュウリのシンプルな
            DATAFORM キュウリとカリカリのベーコンの
            DATAFORM サクサクのチキンカツと甘酸っぱいソースの
            DATAFORM 桜の花の香りが漂う、クリームチーズとベリーの
            DATAFORM さくらんぼジャムと桃のスライスの
            DATAFORM クリームチーズとストロベリージャムの
        ENDDATA
    CASE 1 ;夏
        PRINTDATA
            DATAFORM スパイシーでジューシーなチキンと、コールスローを合わせた
            DATAFORM 甘酸っぱいパイナップルと香ばしいハム、マヨネーズを合わせた南国スタイルの
            DATAFORM スパイシーなチョリソーとサルサソースのホット
            DATAFORM 蟹とレタスをレモン風味のマヨネーズで和えた
            DATAFORM 蟹とアボカドにマヨネーズを合わせた
            DATAFORM トーストしたシンプルなパンに、レタスと、トマトにアボカドをのせてレモンやハーブで味付けした
            DATAFORM カリッと焼かれたベーコンとタルタルソースの
            DATAFORM 冷たいチキンと甘いバーベキューソースの
            DATAFORM クロワッサンにサーモン、オニオン、レタスを挟んだ
            DATAFORM フレッシュなトマトとアボカドに合わせて海老を贅沢に使った
            DATAFORM トマトとモッツァレラチーズにバジルソースの
            DATAFORM フレッシュな海老、アボカド、そしてレモンマヨネーズの
            DATAFORM バーベキューソースにチキンとベーコンを合わせた
            DATAFORM ぎっしりとレタスと照り焼きチキンが挟まれた
            DATAFORM パイナップル、マンゴー、キウイのトロピカルな
        ENDDATA
    CASE 2 ;秋
        PRINTDATA
            DATAFORM ほくほくのかぼちゃとジューシーなチキンを合わせた
            DATAFORM なめらかなカボチャと濃厚なクリームチーズの
            DATAFORM 甘酸っぱいハニーマスタードソースのチキンとにんじんの
            DATAFORM ふんわりとしたパンに、かぼちゃのペーストにローストビーフ、ルッコラを挟んだ
            DATAFORM 甘酸っぱいクランベリーソースとリンゴの
            DATAFORM シナモンとキャラメルソースをかけた洋なしの
            DATAFORM ローストチキンと、クリーミーなチェダーチーズの
            DATAFORM カボチャとサツマイモのマヨネーズ和えにハムを合わせた
            DATAFORM ベーコンときのこたっぷりの
            DATAFORM コンソメ風味の秋鮭とマヨネーズ
            DATAFORM 照り焼きチキンに卵とタルタルソースを合わせた
        ENDDATA
    CASE 3 ;冬
        PRINTDATA
            DATAFORM ジューシーなローストビーフとたまねぎのホット
            DATAFORM にんじんとレタスのキャロットラペ
            DATAFORM 直火焼きローストビーフとマヨネーズの
            DATAFORM ほくほくのポテトと香ばしいベーコンの
            DATAFORM きのことカリカリのベーコンととろけるチーズの
            DATAFORM サワークリームとチキンとキャベツとトマトの
            DATAFORM クリーミーなチーズとカリカリのベーコン
            DATAFORM マスタードとジューシーでスパイシーなソーセージの
            DATAFORM たっぷりチーズとホワイトソースのベーコン
            DATAFORM じゃがいもとローストビーフのトースト
            DATAFORM ベーコンとチェダーチーズととろけるバターの
            DATAFORM ベーコンとチーズのガーリックバター
            DATAFORM スモークサーモンとクリームチーズの
        ENDDATA
ENDSELECT
;～を選んだ、部分
PRINTDATAW
    DATAFORM サンドウィッチを選んだ！
    DATAFORM サンドウィッチにした！
    DATAFORM サンドウィッチを食べることにした！
    DATAFORM サンドウィッチを%ANAME(QST_RANDCHARA())%が選んだ
ENDDATA
PRINTL

;-------------------------------------------------
;EV21- 噴水（会話イベント）
;-------------------------------------------------
;噴水の描写＋会話テキスト予定
;ほのぼのしたクエストだけど、ここでの会話だけちょっと世知辛い感じにしたいかも
@QST_ADD_EVENT_QUEST_19_EVENT_21
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "噴水"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_21
;夜のテキスト→四季
SIF HO_OCLOCK() > 1900
    PRINTFORMW 噴水は月光に照らされている
PRINTL

SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTFORMW 桜の花びらが風に舞っている……
        PRINTFORML 噴水の水しぶきが軽やかに舞い上がっている
    CASE 1 ;夏
        PRINTFORML 太陽の光が眩しい……
        PRINTFORMW 噴水の水しぶきが涼しげだ
    CASE 2 ;秋
        PRINTFORML 落ち葉が風に舞っている……
        PRINTFORMW 噴水の水しぶきが美しい
    CASE 3 ;冬
        PRINTFORML 空気が冷たい……
        PRINTFORMW 噴水は氷に覆われ、静かに凍りついている
ENDSELECT

;-------------------------------------------------
;EV22 噴水前広場
;-------------------------------------------------
;一応広場扱いにしておくけど、テキストのみ(会話主体)でもいいかも
@QST_ADD_EVENT_QUEST_19_EVENT_22
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "噴水前広場"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_22
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL QST_SET_EVENT_SLOT_NAME(, @"広場から出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;擬似的に設備:小舞台(ID=205)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 205
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;メッセージ
    IF HO_OCLOCK() < 1900
        SELECTCASE HO_SEASON()
            CASE 0 ;春
                PRINTFORML 広場には色とりどりの花々が咲き誇り、爽やかな風が吹き抜ける
                PRINTL
                PRINTFORMW 周囲には人々が集まり、仕事や家事の合間で一息つく姿が見受けられた
            CASE 1 ;夏
                PRINTFORMW 日差しが眩しい中、噴水広場には人々が涼を求めて集まっている
            CASE 2 ;秋
                PRINTFORML 広場の周りには木々の葉がひらひらと舞い降りては落ちていく
                PRINTFORML 秋の心地よい風が吹き抜ける
                PRINTL
                PRINTFORMW 広場には一際落ち着いた雰囲気が広がっている
            CASE 3 ;冬
                PRINTFORML 広場は雪で覆われており、真っ白な景色が広がっている
                PRINTFORML 白い息を吐きながら人々は寒さに凍えている
                PRINTL
                PRINTFORMW 雪と寒さのせいか人影はまばらだ
        ENDSELECT
    ELSE
        ;夜のテキスト
        PRINTFORML 広場の周りには人の姿はまばらで、たまに通りかかる人々の足音や会話の声が聞こえるのみだ
        PRINTL
        PRINTFORMW 夜空には星が輝き、月が優しく地上を照らしている
    ENDIF
ELSE
    CALL QST_SET_EVENT_SLOT_NAME(, @"広場")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    PRINTFORMW 広場を離れることにした……
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV23 噴水前のベンチ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_23
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "噴水前のベンチ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_23
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%は噴水前のベンチに腰掛けた
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチを立つ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;擬似的に設備:ベンチ(ID=22)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 22
    WAIT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%はベンチを立ち去った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"噴水前のベンチ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV30 路地裏
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_30
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "路地裏に入る"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_30
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL FL_PAUSEBGM()
    PRINTFORMW %QST_LEADER_NAME()%は人目のつかない路地裏に忍び込んだ……
    CALL QST_QUEST_19_EVENT_SEASON_AIR_TEXT
    CALL QST_SET_EVENT_SLOT_NAME(, @"路地裏を出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
    ;擬似的に設備:路地裏(ID=900)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 900
ELSE
    CALL FL_REPLAYBGM()
    PRINTFORMW %QST_LEADER_NAME()%は路地裏から出た……
    CALL QST_SET_EVENT_SLOT_NAME(, @"路地裏に入る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV40 店を見て時間を潰す
;-------------------------------------------------
;EV40 夜にしたい……とかあるので時間経過イベント
@QST_ADD_EVENT_QUEST_19_EVENT_40
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "店を見て時間を潰す"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_40
PRINTFORML ここにはささやかながら様々な店が並んでいる。
PRINTFORMW 店を見て時間をつぶそうか？
PRINTL
CALL ASK_MULTI_JUDGE("１時間ほど店を見る", 1, "３時間ほど店を見る", 1, "やめておく", 1)
CLEARLINE 1
PRINTL
SELECTCASE RESULT
    CASE 0
        PRINTFORMW 一時間ほど店を見て時間をつぶすことにした……
        PRINTL
        CALL QST_QUEST_19_EVENT_40_RANDOM_TEXT
        HO_TIME += 60
    CASE 1
        PRINTFORMW 三時間ほど店を見て時間をつぶすことにした……
        PRINTL
        CALL QST_QUEST_19_EVENT_40_RANDOM_TEXT
        HO_TIME += 180
    CASEELSE
        PRINTFORMW ……やめておこう
        RETURN
ENDSELECT

;;; - 時間経過時のテキスト　もうちょっと欲しいけど思いつかないのでこんなもんで
;;; - 今はランダム以外の共通部分もつっこんでるけど、よくない
@QST_QUEST_19_EVENT_40_RANDOM_TEXT
PRINTL ……
PRINTDATAW
    DATAFORM ぼんやりと雑貨屋の店内を眺めながら、ゆっくりとした時間を過ごした……
    DATAFORM 店頭の商品を眺め、穏やかな時間を過ごした……
    DATAFORM ……ひとときの休息を楽しんだ
    DATAFORM ……気がついたら予定の時間間際になっていた
    DATAFORM ……ゆっくりとするのも悪くない、と%ANAME(QST_RANDCHARA())%は思った
    DATAFORM 雑貨を眺めることで、気分転換になった……
    DATAFORM 素朴な商店街をぶらつき、店先に並ぶ品々をじっくり眺めた……
    DATAFORM ……時折聞こえる笑い声が、商店街に楽しい雰囲気を与えているように感じた……
    DATAFORM 時間を潰すために、様々な店をぶらついた……
    DATAFORM ……街を彩る様々な店の窓辺には、美しい花が飾られていた
    DATAFORM ゆっくりと歩きながら、心ゆくまで店先を見て回った……
    DATAFORM 時間を忘れて工芸品を眺めた……
    DATAFORM 薬草店では違いのわからないほど様々な薬草が陳列されていた……
    DATAFORM ……古書店で様々な本を見ている内に、あっという間に時間が過ぎた
    DATAFORM 古書店では難解な魔法書が並んでいた……
    DATAFORM 日常品から珍しい品物まで、さまざまなものを眺めた……
    DATAFORM 地元の花屋では、鮮やかな色とりどりの花が並べられていた……
    DATAFORM パン屋には、焼き立てのパンが並んでいて、とても食欲をそそる……
    DATAFORM 異国の土産物屋で、珍しいお土産を眺め、楽しい時間を過ごした……
    DATAFORM 装飾品店では、鳥の羽を使った髪飾りが売っていた……
    DATAFORM 雑貨屋には赤べこが飾られていた……
    DATAFORM ひとつずつ手作りされた木彫りのタリズマンを眺め、その細やかな彫刻に感心した……
    DATAFORM 鞭が売っており、戦いに役立つだろうかと思ったが、『人間用』のようだ……
    DATAFORM 装飾品店では、美しいグラントロワ製の金属細工がケース内に並べられていた……
    DATAFORM アンティーク店ではコレクター用としてディナリア銀貨が売られていた……
ENDDATA
PRINTFORMW ……
PRINTL
PRINTFORMW ……時間のようだ

;-------------------------------------------------
;EV41 花壇横のベンチ
;-------------------------------------------------
;ウフフ禁止。描写盛ってもいいかなあ
@QST_ADD_EVENT_QUEST_19_EVENT_41
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "花壇横のベンチ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_41
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%は花壇横のベンチに腰掛けた
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチを立つ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;擬似的に設備:ベンチ(ID=22)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 22
    WAIT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%はベンチを立ち去った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"花壇横のベンチ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
ENDIF

;-------------------------------------------------
;EV42 小さな店
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_42
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "メイポール雑貨屋"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_42
SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTFORML メイポール雑貨屋の店先には植木鉢が置かれ、ささやかながら花が植えられている
    CASE 1 ;夏
        PRINTFORML メイポール雑貨屋の店先は鮮やかな緑に包まれ、植木鉢に植えられた花々が色とりどりの花を咲かせている
    CASE 2 ;秋
        PRINTFORML メイポール雑貨屋の店先には植木鉢が置かれ、色鮮やかなダリアが植えられている
    CASE 3 ;冬
        PRINTFORML 店先には雪が積もり、植木鉢も看板も白い雪に包まれている
ENDSELECT
PRINTW
PRINTFORML 店内は狭く、木造建築特有の湿った木の匂いが立ち込めている
PRINTFORML カウンターや棚は長年使われた形跡があり年季を感じさせる
PRINTL
PRINTFORML 棚にはパンや動物の卵が並び、横にはには蒸留酒の瓶がずらりと並ぶ
PRINTFORML 奥には木材等資材も積まれている
PRINTL
PRINTFORML カウンターの上には赤べこが飾られていた
PRINTW
PRINTFORMW 「いらっしゃい！　何かお求めですか？」
PRINTL
CALL HO_SHOP_MENU("QUEST_19_メイポール雑貨屋", "メイポール雑貨屋")
PRINTL
PRINTFORMW ……メイポール雑貨屋を離れることにした

;-------------------------------------------------
;EV43 雑貨屋（冒険者っぽいもの。武器も売ってる）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_43
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "新芽の道具屋"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_43
PRINTFORML 新芽の道具屋は堅牢な石造りの建物で、その軒先には看板が掲げられている
SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTFORML 春の陽光が店の看板を照らしている
    CASE 1 ;夏
        PRINTFORML 石造りの建物は夏の日差しを受け、涼しげな影を落としている
    CASE 2 ;秋
        PRINTFORML 店に入ると、そこには何人かの冒険者と思われる人々が品物を見ている
    CASE 3 ;冬
        PRINTFORML 暖炉には火が焚かれ、程良く暖かな室内で多くの冒険者達が品物を見ている
ENDSELECT
PRINTFORML 棚にはランタンや薬草、ポーションにちょっとした調理器具など冒険者向けの武器や道具が陳列されていた
PRINTFORML 壁には短剣や縄が掛けられ、カウンターには駆け出し冒険者向けのガイド本も置かれている
PRINTW
PRINTFORMW 「いらっしゃい、冒険のお供にぴったりのアイテムが揃っているよ」
PRINTL
CALL HO_SHOP_MENU("QUEST_19_新芽の道具屋", "新芽の道具屋")
PRINTL
PRINTFORMW ……新芽の道具屋を離れることにした

;-------------------------------------------------
;EV44 服飾店　白鹿
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_44
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "服飾店　白鹿"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_44
PRINTFORML 店内に入ると店員がこちらに顔を向け、落ち着いた声で「いらっしゃいませ」と会釈をする
PRINTL
PRINTFORML 白を基調とした清潔感のある店内には、布地や洋服が丁寧に陳列されている
IF HO_OCLOCK() < 1800
    SIF HO_SEASON() != HO_季節_夏
        PRINTFORML 窓からは柔らかな光が差し込み、店内を明るく照らしている
    SIF HO_SEASON() == HO_季節_夏
        PRINTFORML 窓からはまぶしいほどの光が差し込み、冷えた氷水が入った水差しがカウンターに置かれている
ELSE
    ;夜のテキスト
    PRINTFORML 店内は暖かな灯りに包まれている
ENDIF
PRINTW
SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTFORML 明るい色合いのワンピースやブラウスが並び、目を引く
    CASE 1 ;夏
        PRINTFORML 上品なデザインのワンピースやブラウスが目を引く
    CASE 2 ;秋
        PRINTFORML 店内の上品なデザインの秋色の服が目を引く
    CASE 3 ;冬
        PRINTFORML 暖かそうな厚手のコートが目を引く
ENDSELECT
PRINTFORML 素材も上等なものを使っているのか、とても品質がよさそうだ
PRINTW
PRINTFORML 「当店では上質な布地や洗練された服を取り揃えております。
PRINTFORMW お客様のお好みに合わせてお選びください」
PRINTL
CALL HO_SHOP_MENU("QUEST_19_服飾店白鹿", "服飾店　白鹿")
PRINTL
PRINTFORMW ……服飾店　白鹿を離れることにした

;-------------------------------------------------
;EV50 商品を購入する（怪しい店）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_50
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "商品を購入する"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_50
CALL HO_SHOP_MENU("QUEST_19_酔狂の店", "酔狂の店")

;-------------------------------------------------
;EV51- 棚の宝石を見る（テキストイベント）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_51
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "棚の宝石を見る"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_51
#DIM CONST EVENT_ID = 51
#DIM CONST RELIC_PRICE_51 = 20000

PRINTFORMW 黒い棚の上、ガラスケースの中に大きな宝石が並べられている
PRINTL
PRINTFORML カット済みにもかかわらず、明らかに輝きが鈍い……
PRINTFORML あまり価値のあるものではなさそうだ
PRINTL
PRINTFORMW せいぜい、河原で見つけてきた少し変わった石と言ったところだろうか
CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
IF ABL:QST_LEADER_ID():知識 > 80 && RESULT == 0
    PRINTFORMW しかし、%QST_LEADER_NAME()%はその石に何かを感じ取ることができた
    PRINTL
    PRINTFORML 値段は%TOSTR(RELIC_PRICE_51, "###,###,##0")%ディナリと書かれているが、果たしてその価値は……？
    PRINTL

    CALL ASK_MULTI_JUDGE("購入する", MONEY >= RELIC_PRICE_51, "購入しない", 1)
    IF RESULT == 0 && MONEY >= RELIC_PRICE_51
        PRINTFORMW %QST_LEADER_NAME()%はその石を購入した
        CALL QST_CHANGE_ITEM(@"ディナリ,-{RELIC_PRICE_51},")
        PRINTL
        ;レリック:古代人の守護石
        {
            CALL QST_GET_WEAPON(
                WPN_装備種別_レリック
                ,"古代人の守護石"
                ,WPN_材質_鉄
                ,0
                ,""
                ,""
                ,WPN_装備種別_レリック
                ,1
                )
        }
        CALL QST_SET_SV_EVENT_FLAG(EVENT_ID, 1)
    ELSE
        PRINTFORMW %QST_LEADER_NAME()%はその石を購入しなかった
    ENDIF
ENDIF

;-------------------------------------------------
;EV52- 壁を見る（テキストイベント）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_52
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "壁を見る"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_52
PRINTFORML ……壁にはさまざまな品々が飾られている
PRINTFORML 仮面が不気味な微笑みを浮かべ、陶芸品は磨き上げられている
PRINTFORMW 地図は古びた色合いで、首飾りは何か不思議な輝きを持っている
PRINTL
PRINTFORML しかし、これらの品々は全て価格が常識を逸脱し、奇妙なほど高額である
PRINTFORMW この品々が本当にその価格に見合うものなのかは疑問だ……

;-------------------------------------------------
;EV53- 悩む女性（会話イベント）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_53
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "悩む女性"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_53
SETCOLOR カラー_ライトブルー
PRINTFORML 「うーん……」
PRINTFORMW 「プレゼントに悩んでこの店に来てみたけど、他の店で探したほうが良さそう……」
SIF HO_SEASON() == HO_季節_夏
    PRINTFORMW 「でも外出ると暑いしなあ……」
SIF HO_SEASON() == HO_季節_冬
    PRINTFORMW 「でも外出ると寒いしなあ……」
RESETCOLOR

;-------------------------------------------------
;EV54- 商品を見る男性（会話イベント）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_54
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "商品を見る男性"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_54
SETCOLOR カラー_ライトブルー
PRINTFORML 「ここで売ってるのは、なんというか……ジョークグッズとでも言えばいいのか？」
PRINTFORMW 「……値段は冗談じゃないが」
PRINTFORMW 「『おもちゃ』とかは、少々遠くてもフォルメンとかに行った方がよっぽど安く手に入る」
RESETCOLOR

;-------------------------------------------------
;EV59 酔狂の店入店時
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_59
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_19_EVENT_59
PRINTFORMW 扉を押し開けて中に入ると、中は暗く埃っぽい空気で満たされていた。
PRINTL
PRINTFORML 所狭しと陳列されている品々は奇妙なものばかりで、怪しい雰囲気が漂っている
PRINTFORML カウンターの奥にもさらによくわからない品々が並んでいる……
PRINTW

;-------------------------------------------------
;EV60 ベンチ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_60
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ベンチ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_60
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%はベンチに腰掛けた
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチを立つ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;擬似的に設備:ベンチ(ID=22)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 22
    ;夜ならウフフ系コマンドの実行を許可
    IF HO_OCLOCK() >= 1900
        FLAG:クエスト_ウフフ許可 = 1
        PRINTFORML 辺りには誰も居ない……
    ENDIF
    WAIT
    CALL QST_QUEST_19_EVENT_SEASON_AIR_TEXT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%はベンチを立ち去った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV61 広場
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_61
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "広場"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_61
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL QST_SET_EVENT_SLOT_NAME(, @"広場から出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;擬似的に設備:広場(ID=21)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 21
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    PRINTFORML 公園の広場へと出た……
    SIF HO_OCLOCK() > 1900
        PRINTFORML 夜の公園はとても静かだ……
    WAIT
ELSE
    CALL QST_SET_EVENT_SLOT_NAME(, @"広場")
    PRINTFORML 公園の広場から出た……
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV62 公衆トイレ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_62
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "公衆トイレ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_62
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT
IF !EVENT_FLAG
    CALL FL_PAUSEBGM()
    PRINTFORMW %QST_LEADER_NAME()%は公衆トイレに入った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"公衆トイレを出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
    ;擬似的に設備:トイレ(ID=9)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 9
ELSE
    CALL FL_REPLAYBGM()
    PRINTFORMW %QST_LEADER_NAME()%は公衆トイレから出た……
    CALL QST_SET_EVENT_SLOT_NAME(, @"公衆トイレ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV71 茂み
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_71
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "茂み"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_71
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL FL_PAUSEBGM()
    PRINTDATAW
        DATAFORM %QST_LEADER_NAME()%は%SPLIT_RAND("静かに/ひそやかに",1)%人目のつかない茂みに忍び込んだ……
        DATAFORM %QST_LEADER_NAME()%は%SPLIT_RAND("忍び足で/静かに/ひそやかに//忍び込むように",1)%人目のつかない茂みの中に足を踏み入れた……
        DATAFORM %QST_LEADER_NAME()%は%SPLIT_RAND("忍び足で/静かに/ひそやかに//忍び込むように",1)%人目のつかない茂みに侵入した……
    ENDDATA
    PRINTL
    CALL QST_QUEST_19_EVENT_SEASON_AIR_TEXT
    CALL QST_SET_EVENT_SLOT_NAME(, @"茂みを出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
    ;擬似的に設備:茂み(ID=12)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 12
ELSE
    CALL FL_REPLAYBGM()
    PRINTFORMW %QST_LEADER_NAME()%は茂みから出た……
    CALL QST_SET_EVENT_SLOT_NAME(, @"茂み")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV72 ベンチ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_72
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ベンチ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_72
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%はベンチに腰掛けた
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチを立つ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
    ;擬似的に設備:ベンチ(ID=22)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 22
    PRINTFORML この辺りには誰もいないようだ……
    WAIT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%はベンチを立ち去った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV73 木々の間
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_19_EVENT_73
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "木々の間"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_73
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL FL_PAUSEBGM()
    PRINTDATAW
        DATAFORM %QST_LEADER_NAME()%は%SPLIT_RAND("静かに/ひそやかに",1)%人目のつかない木々の間に忍び込んだ……
        DATAFORM %QST_LEADER_NAME()%は%SPLIT_RAND("忍び足で/静かに/ひそやかに//忍び込むように",1)%人目のつかない木々の間に足を踏み入れた……
        DATAFORM %QST_LEADER_NAME()%は%SPLIT_RAND("忍び足で/静かに/ひそやかに//忍び込むように",1)%人目のつかない木々の間に侵入した……
    ENDDATA
    PRINTL
    CALL QST_QUEST_19_EVENT_BUSH_TEXT
    CALL QST_QUEST_19_EVENT_SEASON_AIR_TEXT
    CALL QST_SET_EVENT_SLOT_NAME(, @"木々の間から出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
    ;擬似的に設備:茂み(ID=12)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 12
ELSE
    CALL FL_REPLAYBGM()
    PRINTFORMW %QST_LEADER_NAME()%は木々の間から出た……
    CALL QST_SET_EVENT_SLOT_NAME(, @"木々の間")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
    ;LOG更新
    CALL QST_QUEST_19_EVENT_0
ENDIF

;-------------------------------------------------
;EV31- 吟遊詩人（会話イベント）
;-------------------------------------------------
; 31下手な吟遊詩人
; 最初の会話だけこのメッセージ、みたいなことしたいけどとりあえずこのままで

@QST_ADD_EVENT_QUEST_19_EVENT_31
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "吟遊詩人"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_19_EVENT_31
SETCOLOR カラー_ライトブルー
PRINTFORMW 「♪ごきげんよう、吟遊詩人の名を持つ者よ～」
PRINTL
PRINTFORML 「……あ、いや、♪私が、吟遊詩人の名を持つものです～」
PRINTFORML 「♪ええと～、音楽と音の力で……伝説を……歌で……届けさせて欲しい～」
RESETCOLOR
PRINTL
PRINTFORMW  ……あまり慣れていない吟遊詩人のようだ。
PRINTL
; ランダム仮実装 はずれでるとわけわからんので３回くらい出しとこう
SETCOLOR カラー_ライトブルー
REPEAT 3
    PRINTFORM ♪
    CALL QST_QUEST_19_EVENT_31_RANDOM_TEXT_5
    PRINTFORMW
    CALL QST_QUEST_19_EVENT_31_RANDOM_TEXT_7
    PRINTFORMW
    CALL QST_QUEST_19_EVENT_31_RANDOM_TEXT_5
    PRINTW
    PRINTL
REND
PRINTFORML ……どうする？
PRINTL
CALL ASK_MULTI_JUDGE("500ディナリを投げる", 1, "放っておく",1)
CLEARLINE 1
SELECTCASE RESULT
    CASE 0
        IF MONEY < 500
            PRINTFORMW %ANAME(QST_RANDCHARA())%はそんなにお金の余裕がないことに気が付いた
            RETURN
        ENDIF
        CALL FL_PLAYSE("Money_0.mp3")
        PRINTFORMW 500ディナリを吟遊詩人へと投げた！
        MONEY -= 500
        PRINTL
        SETCOLOR カラー_ライトブルー
        PRINTDATAL
            DATAFORM 「え！？　あ、ありがとうございます！」
            DATAFORM 「ああ、これはなんと太っ腹な方！感謝いたします！」
            DATAFORM 「あっ、お金！　吟遊詩人として認めていただき、ありがとうございます！」
            DATAFORM 「な、なんと…！　ありがとうございます！」
            DATAFORM 「私の歌が、あなたの心に響いたのだろうか……、感謝します！」
            DATAFORM 「あ、ありがとう！　♪この気前の良さ、まるで太陽のように暖かい！」
            DATAFORM 「♪あなたの寛大な心が、私の心に光をもたらす～ありがとう～」
        ENDDATA
        RESETCOLOR
        PRINTL
        PRINTFORMW 吟遊詩人は驚き、喜んでいる……
    CASEELSE
        PRINTFORMW そっとしておこう……
        RETURN
ENDSELECT

;-------------------------------------------------
;EV31- 吟遊詩人-DATASET
;-------------------------------------------------
; 乱数で面白歌つくるとかそういう方向でもいいかなと思ったので仮実装 575で

;;; 約５文字/吟遊詩人のランダム生成メッセージ
@QST_QUEST_19_EVENT_31_RANDOM_TEXT_5
PRINTDATA
    ; スキル
    DATAFORM 軽減値
    DATAFORM 行動力
    DATAFORM エトワール
    DATAFORM ミダスの手
    DATAFORM マナボルト
    DATAFORM 詠唱3
    DATAFORM 歌唱1
    DATAFORM デコイ4
    DATAFORM 剣の舞
    ;アイテム
    DATAFORM 薬草と
    DATAFORM ポーションと
    DATAFORM 鉄鉱石
    DATAFORM 銀鉱石
    DATAFORM 金鉱石
    DATAFORM マナ結晶
    DATAFORM インゴット
    DATAFORM マナグラス
    DATAFORM 蒸留酒
    ; 武器
    DATAFORM マスケット
    ; レリック
    ; その他
    DATAFORM 100ディナリ
    DATAFORM 1000ディナリ
    DATAFORM 名誉点
    DATAFORM 淫堕点
    DATAFORM 治療点
    ; クラフト
    DATAFORM 領内の
    DATAFORM 娯楽室
    DATAFORM 家畜小屋
    DATAFORM ちゃぶ台と
    DATAFORM 露天風呂
    DATAFORM ラージソファ
    DATAFORM 作業台
    DATAFORM 錬金台
    DATAFORM 蓄音機
    DATAFORM 大居間で
    ; クエスト
    DATAFORM ゴブリンの
    DATAFORM フォルメンの
    DATAFORM 炭鉱街
    DATAFORM 深い森
    DATAFORM アマゾネス
    DATAFORM メスガキの
    DATAFORM ナイトメア
    DATAFORM オートマタ
    DATAFORM ゴブリン達
    DATAFORM スケルトン
    DATAFORM ワイバーン
    DATAFORM 死の影が
    DATAFORM 盗賊団
    DATAFORM 泥棒猫
    DATAFORM トネリコの
    DATAFORM 海岸の
    DATAFORM 水遊び
    DATAFORM 難破船
    ; 設定
    DATAFORM 貴族たち
    DATAFORM 錬金術
    DATAFORM 冒険者
    DATAFORM 古帝国
    DATAFORM ノルディアの
    DATAFORM アカデミー
    DATAFORM ノアレルム
    DATAFORM へスペリア
    DATAFORM ローレリアム
    DATAFORM 王党派
    DATAFORM フロリアの
    ; era
    DATAFORM フェイスハガー
    DATAFORM あなただけ
    DATAFORM 搾乳機
    DATAFORM 倒錯度
    DATAFORM 欲望の
    DATAFORM 未経験
    DATAFORM 合意済
    DATAFORM 女騎士
    DATAFORM 太い脚
    DATAFORM ざこざこの❤
    DATAFORM ざぁこざこ❤
    DATAFORM ❤❤❤❤❤
    DATAFORM しっぽりと
    DATAFORM 路地裏で
    DATAFORM 二人きり
    DATAFORM 机下
    ; その他
    DATAFORM 潜むもの
    DATAFORM 憧れる
    DATAFORM 素晴らしい
    DATAFORM 最高だ
    DATAFORM 最強だ
    DATAFORM 大好きです
    DATAFORM 今流行り
    DATAFORM 一番の
    DATAFORM 禁断の
    DATAFORM 大量だ
    DATAFORM 今日だけは
    DATAFORM ありがとう
    DATAFORM 疎外感
    DATAFORM こっそりと
    DATAFORM 集めよう
ENDDATA

;;; 約７文字/吟遊詩人のランダム生成メッセージ
@QST_QUEST_19_EVENT_31_RANDOM_TEXT_7
PRINTDATA
    ; スキル
    DATAFORM アーマーピアサー
    DATAFORM ポイズンチェイサー
    DATAFORM 戦術詠唱
    DATAFORM ミスディレクション
    DATAFORM 致命の一撃
    DATAFORM 牽制攻撃
    DATAFORM 前衛防御
    DATAFORM スケープゴート
    DATAFORM ディミヌエンドの
    DATAFORM マナストームが
    DATAFORM キャストオフが
    DATAFORM キャストオフで
    ; アイテム
    DATAFORM 技巧の宝玉
    DATAFORM 淫堕の宝玉
    DATAFORM 栄誉勲章
    DATAFORM 銀インゴット
    DATAFORM 金インゴット
    DATAFORM 魔獣の肉片
    DATAFORM 魔獣の心核
    DATAFORM 魔導触媒
    DATAFORM 金属細工の
    DATAFORM メイガスの夜
    DATAFORM ディナリア銀貨
    DATAFORM 無垢のスコーン
    DATAFORM キルケの小箱
    DATAFORM 建築資材
    ; 武器
    DATAFORM 鉄の短剣
    DATAFORM 銀の短剣
    DATAFORM 鉄の戦槌
    DATAFORM 鉄のレイピア
    ; レリック
    DATAFORM 赤べこ +3
    DATAFORM 淫魔の導印
    DATAFORM 淫魔の首枷
    ; クラフト
    DATAFORM ラウンドテーブル
    DATAFORM ティーテーブルが
    DATAFORM 木陰の水辺で
    ; クエスト
    DATAFORM 夕暮港
    DATAFORM 隊商護衛は
    DATAFORM 隊商護衛の
    DATAFORM 無限に続く
    DATAFORM エメリア大学
    DATAFORM 自動人形
    DATAFORM 触手の怪物
    DATAFORM 秘密の迷宮
    DATAFORM 次元の渦が
    DATAFORM デミスキュレー
    ; 設定
    DATAFORM コモンウェルスの
    DATAFORM リナシメントは
    DATAFORM 爆発卵
    DATAFORM 中央世界
    DATAFORM レルム半島
    ; era
    DATAFORM 一線超えない
    DATAFORM あなた以外の
    DATAFORM なんと３人
    DATAFORM 露出してる
    DATAFORM 不思議な体質
    DATAFORM 震える玩具
    DATAFORM 代わる代わるの
    DATAFORM 奉仕経験
    DATAFORM 成人向けの
    DATAFORM ホブゴブリンは
    DATAFORM ねぶりスライム
    DATAFORM ❤❤中毒
    ; その他
    DATAFORM めぐる季節は
    DATAFORM 掛け合わせるは
ENDDATA

;-------------------------------------------------
;情景描写テキスト
;-------------------------------------------------
;;;日差しと風と空気感のテキスト
@QST_QUEST_19_EVENT_SEASON_AIR_TEXT
IF HO_OCLOCK() < 1600
    ; 16:00までは朝・昼
    SELECTCASE HO_SEASON()
        CASE 0 ;春
            PRINTDATAW
                DATAFORM %SPLIT_RAND("春の太陽の光/太陽の光/日差し/柔らかな光",1)%が%SPLIT_RAND("優しく/穏やかに",1)%降り注いでいる……
                DATAFORM %SPLIT_RAND("春の太陽の光/太陽の光/日差し/柔らかな光",1)%が辺りを%SPLIT_RAND("優しく/穏やかに",1)%照らしている……
                DATAFORM %SPLIT_RAND("春の太陽の光/太陽の光/柔らかな光",1)%が降り注いでおり、穏やかな暖かさだ……
                DATAFORM %SPLIT_RAND("春の太陽の光/太陽の光/日差し",1)%は穏やかだ……
                DATAFORM 朝の日差しは柔らかく、穏やかな光が地面を照らしている……
                DATAFORM 春の陽気が暖かく、心地よい……
            ENDDATA
        CASE 1 ;夏
            PRINTDATAW
                DATAFORM 太陽の光が%QST_LEADER_NAME()%を%SPLIT_RAND("照らす/照りつけている/じりじりと照りつけている/強く照らしている",1)%……
                DATAFORM 太陽の光が%SPLIT_RAND("照らす/まぶしく降り注いでいる/照りつけている/強く照らしている/強く降り注いでいる",1)%……
                DATAFORM 太陽の熱気が辺り%SPLIT_RAND("一面に広がって/を包んで/を満たして",1)%いる……
                DATAFORM %SPLIT_RAND("夏の日中の光/真夏の日差し/太陽の光",1)%は眩しいほどに強い……
                DATAFORM %SPLIT_RAND("夏の日中の光/真夏の日差し/太陽の光",1)%は強烈で、空気が蒸し暑い……
                DATAFORM 夏の熱気は汗ばむような暑さだ……
                DATAFORM 汗が%QST_LEADER_NAME()%の額を伝う……
            ENDDATA
        CASE 2 ;秋
            PRINTDATAW
                DATAFORM %SPLIT_RAND("そよ風/北風/寒風/秋の風/木枯らし/風",1)%が%QST_LEADER_NAME()%の周りを吹き抜けた……
                DATAFORM %SPLIT_RAND("そよ風/北風/寒風/秋の風/木枯らし/風",1)%が%SPLIT_RAND("吹いている/涼しい/少し冷たい/少し肌寒い",1)%……
                DATAFORM %SPLIT_RAND("そよ風/北風/寒風/秋の風/木枯らし/風",1)%が寂しげに吹いている……
                DATAFORM %SPLIT_RAND("秋の日差し/秋の朝の太陽の光/太陽の光/日差し",1)%は%SPLIT_RAND("淡く/穏やかで",1)%、%SPLIT_RAND("優しい光が地面を照らしている/木々に深い影を落としている",1)%……
                DATAFORM %SPLIT_RAND("秋の太陽の光/太陽の光/日差し/柔らかな光",1)%が%SPLIT_RAND("淡く/優しく/穏やかに",1)%%SPLIT_RAND("降り注いでいる/差し込んでいる",1)%……
                DATAFORM %SPLIT_RAND("秋の太陽の光/太陽の光/日差し/柔らかな光",1)%が辺りを優しく照らしている……
            ENDDATA
        CASE 3 ;冬
            PRINTDATAW
                DATAFORM 冷たい風が%SPLIT_RAND("吹き抜ける/頬を撫でる",1)%……
                DATAFORM 空気%SPLIT_RAND("が冷たく澄んでいる/が冷たい/は刺すように冷たい",1)%……
                DATAFORM 太陽の光が淡く照りつけている……
                DATAFORM 空気が冷たく、%SPLIT_RAND("吐く息は白く染まっている/身体の芯まで凍えるような寒さだ",1)%……
                DATAFORM %SPLIT_RAND("冬の日差し/日差し/太陽の日差し",1)%は弱く、寒さが%SPLIT_RAND("身にしみる/厳しい",1)%……
                DATAFORM %SPLIT_RAND("冬の日差し/日差し/太陽の日差し",1)%は淡く、体の芯まで冷える寒さを感じる……
            ENDDATA
    ENDSELECT
ELSEIF HO_OCLOCK() < 1900
    ; 19:00までは夕方
    SELECTCASE HO_SEASON()
        CASE 0 ;春
            PRINTDATAW
                DATAFORM 夕方の%SPLIT_RAND("太陽の光/日差し",1)%は柔らかく、優しい色合いを与えている……
                DATAFORM 夕暮れ時の%SPLIT_RAND("太陽の光/日差し",1)%は穏やかだ……
                DATAFORM %SPLIT_RAND("夕方の風/そよ風/夕暮れ時の風",1)%が穏やかに吹いている……
                DATAFORM %SPLIT_RAND("夕方の風/そよ風/夕暮れ時の風",1)%が心地よく吹き抜け、春の香りを運んでくる……
            ENDDATA
        CASE 1 ;夏
            PRINTDATAW
                DATAFORM 夕方の%SPLIT_RAND("太陽の光/日差し",1)%はまだまだ強く、空気が熱く満ちている……
                DATAFORM %SPLIT_RAND("夕暮れ時の太陽の光/夕陽",1)%はまだまだまぶしく、空を赤く染めている……
                DATAFORM 夕方の%SPLIT_RAND("太陽の光/日差し",1)%はじりじりと暑さを感じさせ、まだまだ蒸し暑い……
                DATAFORM %SPLIT_RAND("夕方の風/そよ風/夕暮れ時の風",1)%が熱気を運び、まだまだ暑さを感じさせる……
                DATAFORM %SPLIT_RAND("夕方の風/そよ風/夕暮れ時の風",1)%が熱い風を運び、蒸し暑い夜を告げる……
            ENDDATA
        CASE 2 ;秋
            PRINTDATAW
                DATAFORM 夕方の%SPLIT_RAND("太陽の光/日差し",1)%は柔らかく、木々に深い影を落としている……
                DATAFORM %SPLIT_RAND("夕暮れ時の太陽の光/夕陽/夕焼け",1)%は美しく、空がオレンジ色に染まっている……
                DATAFORM 夕方の%SPLIT_RAND("太陽の光/日差し",1)%は淡く、穏やかな暖かさが広がっている……
                DATAFORM %SPLIT_RAND("そよ風/北風/寒風/秋の風/木枯らし/風/夕暮れ時の風",1)%が寂しげに吹き抜け、秋の深まりを感じさせる……
                DATAFORM %SPLIT_RAND("そよ風/北風/寒風/秋の風/木枯らし/風/夕暮れ時の風",1)%が吹いている……
            ENDDATA
        CASE 3 ;冬
            PRINTDATAW
                DATAFORM 冬の夕陽は淡く、寒さがより一層感じられる……
                DATAFORM 夕焼けの中、冷たい風が吹き抜ける……
                DATAFORM %SPLIT_RAND("夕暮れ時の太陽の光/夕陽",1)%は淡く、冷たい風が吹き抜ける……
                DATAFORM 夕方の%SPLIT_RAND("太陽の光/日差し",1)%は弱く、寒さが身にしみる……
                DATAFORM %SPLIT_RAND("そよ風/北風/寒風/秋の風/木枯らし/夕暮れ時の風",1)%が冷たく吹き抜け、冬の深まりを感じさせる……
                DATAFORM 夕暮れ時の%SPLIT_RAND("北風/寒風/秋の風/木枯らし/風",1)%は凍えるような寒さだ……
            ENDDATA
    ENDSELECT
ELSE
    ; それ以降は夜
    SELECTCASE HO_SEASON()
        CASE 0 ;春
            PRINTDATAW
                DATAFORM 春の夜は静かで%SPLIT_RAND("空/星空/星",1)%が美しい……
                DATAFORM 淡い月明かりが%QST_LEADER_NAME()%を優しく照らしている……
                DATAFORM %SPLIT_RAND("風/そよ風",1)%が穏やかに吹いている……
                DATAFORM 春の夜は寒さを感じさせない、穏やかな温かさだ……
                DATAFORM 春の夜は穏やかで、月明かりが優しく照らしている……
            ENDDATA
        CASE 1 ;夏
            PRINTDATAW
                DATAFORM 夏の夜の%SPLIT_RAND("空/星空/夜空",1)%は輝く星が美しい……
                DATAFORM 夏の夜は蝉の鳴き声が静かに響き、夜風が涼しい……
                DATAFORM 月明かりが地面を照らしている……
                DATAFORM 熱気が辺り%SPLIT_RAND("一面に広がって/を包んで/を満たして",1)%いる……
                DATAFORM 夏の夜の空気が蒸し暑い……
                DATAFORM 夏の熱気は汗ばむような暑さだ……
                DATAFORM 汗が%QST_LEADER_NAME()%の額を伝う……
            ENDDATA
        CASE 2 ;秋
            PRINTDATAW
                DATAFORM 秋の夜の%SPLIT_RAND("空/星空/夜空",1)%は澄んでおり、星が美しく輝く……
                DATAFORM 淡い月明かりが地面を照らし、静寂が辺りを包んでいる……
                DATAFORM 秋の夜は風が凪ぎ、静かで穏やかだ……
                DATAFORM 冷たい風が吹き抜け、深い静寂が広がる……
                DATAFORM 秋の夜は静かで、星がきらめく夜空が美しい……
            ENDDATA
        CASE 3 ;冬
            PRINTDATAW
                DATAFORM 冬の夜の%SPLIT_RAND("空/星空/夜空",1)%は深く澄み、星がきらめいている……
                DATAFORM %SPLIT_RAND("冬夜/寒夜/冬の夜/真冬の夜",1)%は%SPLIT_RAND("寒さが身にしみる/凍えるような寒さだ/冷たく静かだ/空気が冷たく澄んでいる/寒さが厳しい",1)%……
                DATAFORM %SPLIT_RAND("寒気/冷気/しんしんと冷える空気/冷たい夜風/凍りつくような空気/冷たい空気",1)%が体に触れる……
            ENDDATA
    ENDSELECT
ENDIF

@QST_QUEST_19_EVENT_BUSH_TEXT
;木々のテキスト
SELECTCASE HO_SEASON()
    CASE 0 ;春
        PRINTDATAW
            DATAFORM 春の風が吹き抜けていく
            DATAFORM 新緑が静かに揺れている
            DATAFORM 風にそよぐ若葉の音が聞こえる
            DATAFORM 鳥たちのさえずりが聞こえる
            DATAFORM 青々とした新芽が生い茂っている
            DATAFORM 草木の香りが漂ってくる
        ENDDATA
    CASE 1 ;夏
        PRINTDATAW
            DATAFORM 茂みの奥に足を踏み入れると、緑濃い葉陰が%QST_LEADER_NAME()%を迎える
            DATAFORM 虫の音が聞こえる
            DATAFORM 緑の葉が風に揺れている
            DATAFORM 草木が緑の葉を揺らしている
            DATAFORM 茂みを照らし、蒸し暑さが広がる
            DATAFORM 草の匂いと湿気を感じる
        ENDDATA
    CASE 2 ;秋
        PRINTDATAW
            DATAFORM 枯葉が足元でサクサクと音を立てる
            DATAFORM 落ち葉が舞っている
            DATAFORM 色づいた葉っぱが風に揺れている
        ENDDATA
    CASE 3 ;冬
        PRINTDATAW
            DATAFORM 雪を踏む%QST_LEADER_NAME()%の足音が響く
            DATAFORM 木々に雪化粧が施されている
            DATAFORM 木々は葉を落とし、どこか寂しげな風景だ
            DATAFORM 枝に雪が降り積もっている
            DATAFORM 木々の枝が白い雪で覆われている
            DATAFORM 足元は雪に覆われており、一歩進むごとにザクザクという音が響く
        ENDDATA
ENDSELECT

;-------------------------------------------------
;ショップの定義
;関数名は@HO_SHOP_MANAGE_SHOP_{ショップ名称}で定義
;-------------------------------------------------

@HO_SHOP_MANAGE_SHOP_QUEST_19_テスト店(DICS_SHOP_STOCK)
#DIMS REF DICS_SHOP_STOCK
DICS_SHOP_STOCK:0 '= @"[ITEM_ID:{ITEM_媚薬}][ITEM_PRICE:4000]"
DICS_SHOP_STOCK:1 '= @"[ITEM_ID:{ITEM_鞭}][ITEM_PRICE:6000]"

@HO_SHOP_MANAGE_SHOP_QUEST_19_メイポール雑貨屋(DICS_SHOP_STOCK)
;日常っぽいもの
#DIMS REF DICS_SHOP_STOCK
DICS_SHOP_STOCK:0 '= @"[ITEM_ID:{ITEM_パン}][ITEM_PRICE:1000]"
DICS_SHOP_STOCK:1 '= @"[ITEM_ID:{ITEM_動物の卵}][ITEM_PRICE:2000]"
DICS_SHOP_STOCK:2 '= @"[ITEM_ID:{ITEM_蒸留酒}][ITEM_PRICE:2000]"
DICS_SHOP_STOCK:3 '= @"[ITEM_ID:{ITEM_木材}][ITEM_PRICE:1000]"
DICS_SHOP_STOCK:4 '= @"[ITEM_ID:{ITEM_鉄インゴット}][ITEM_PRICE:1300]"

@HO_SHOP_MANAGE_SHOP_QUEST_19_新芽の道具屋(DICS_SHOP_STOCK)
;冒険者っぽいもの
#DIMS REF DICS_SHOP_STOCK
DICS_SHOP_STOCK:0 '= @"[ITEM_ID:{ITEM_薬草}][ITEM_PRICE:400]"
DICS_SHOP_STOCK:1 '= @"[ITEM_ID:{ITEM_ポーション}][ITEM_PRICE:1500]"
DICS_SHOP_STOCK:2 '= @"[ITEM_ID:{ITEM_縄}][ITEM_PRICE:5000]"
DICS_SHOP_STOCK:3 '= @"[WEAPON_ID:1][EQUIP_TYPE:0][WEAPON_NAME:鉄のナイフ][WEAPON_RANK:0][MATERIAL:{WPN_材質_鉄}][ITEM_PRICE:8000]"

@HO_SHOP_MANAGE_SHOP_QUEST_19_服飾店白鹿(DICS_SHOP_STOCK)
;デフォルト一般人服の範囲とすこし クラフトの楽しみを奪わない程度に！
#DIMS REF DICS_SHOP_STOCK
DICS_SHOP_STOCK:0 '= @"[ITEM_ID:{ITEM_布}][ITEM_PRICE:6000]"
DICS_SHOP_STOCK:1 '= @"[CLOTH_ID:12][CLOTH_PART:{衣装部位_上服}][ITEM_PRICE:8000]" ;上服_ブラウス 8000
DICS_SHOP_STOCK:2 '= @"[CLOTH_ID:13][CLOTH_PART:{衣装部位_上服}][ITEM_PRICE:8000]" ;上服_ワイシャツ 8000
DICS_SHOP_STOCK:3 '= @"[CLOTH_ID:5][CLOTH_PART:{衣装部位_下服}][ITEM_PRICE:8000]" ;下服_ズボン 8000
DICS_SHOP_STOCK:4 '= @"[CLOTH_ID:4][CLOTH_PART:{衣装部位_下服}][ITEM_PRICE:8000]" ;下服_スカート 8000
DICS_SHOP_STOCK:5 '= @"[CLOTH_ID:19][CLOTH_PART:{衣装部位_上下服}][ITEM_PRICE:10000]" ;上下服_ワンピース 10000
DICS_SHOP_STOCK:6 '= @"[CLOTH_ID:13][CLOTH_PART:{衣装部位_下着上}][ITEM_PRICE:4000]" ;下着上_ブラ 4000
DICS_SHOP_STOCK:7 '= @"[CLOTH_ID:11][CLOTH_PART:{衣装部位_下着下}][ITEM_PRICE:4000]" ;下着下_パンティ 4000
DICS_SHOP_STOCK:8 '= @"[CLOTH_ID:8][CLOTH_PART:{衣装部位_下着下}][ITEM_PRICE:4000]" ;下着下_トランクス
DICS_SHOP_STOCK:9 '= @"[CLOTH_ID:11][CLOTH_PART:{衣装部位_靴下}][ITEM_PRICE:2000]" ;靴下_2000
DICS_SHOP_STOCK:10 '= @"[CLOTH_ID:19][CLOTH_PART:{衣装部位_靴}][ITEM_PRICE:5000]" ;靴_5000
SIF HO_SEASON() == HO_季節_春 || HO_季節_夏
DICS_SHOP_STOCK:11 '= @"[CLOTH_ID:14][CLOTH_PART:{衣装部位_上着}][ITEM_PRICE:8000]" ;上着_ベスト_8000
SIF HO_SEASON() == HO_季節_秋 || HO_季節_冬
DICS_SHOP_STOCK:11 '= @"[CLOTH_ID:5][CLOTH_PART:{衣装部位_上着}][ITEM_PRICE:10000]" ;上着_コート_10000

@HO_SHOP_MANAGE_SHOP_QUEST_19_酔狂の店(DICS_SHOP_STOCK)
;変な店。とにかく高い
#DIMS REF DICS_SHOP_STOCK
DICS_SHOP_STOCK:0 '= @"[ITEM_ID:{ITEM_骨}][ITEM_PRICE:10000]"
DICS_SHOP_STOCK:1 '= @"[ITEM_ID:{ITEM_ディナリア銀貨}][ITEM_PRICE:50000]"
DICS_SHOP_STOCK:2 '= @"[ITEM_ID:{ITEM_媚薬}][ITEM_PRICE:10000]"
DICS_SHOP_STOCK:3 '= @"[ITEM_ID:{ITEM_バイブ}][ITEM_PRICE:30000]"

;-------------------------------------------------
;EV1- テスト
;-------------------------------------------------
;EV1 てすといべんと
; @QST_ADD_EVENT_QUEST_19_EVENT_1
; #DIMS EVENT_TRIGGER = "EVENT_SLOT"
; #DIMS EVENT_SLOT_NAME = "テスト"
; RESULTS:0 = %EVENT_TRIGGER%
; RESULTS:1 = %EVENT_SLOT_NAME%

; @QST_QUEST_19_EVENT_1
;-------------------------------------------------
;以下未実装やメモなど
;-------------------------------------------------
; ◇テキスト切り出して関数にしたものリスト
; QST_QUEST_19_EVENT_20_SEASON_SANDWICH
; QST_QUEST_19_EVENT_40_RANDOM_TEXT
; QST_QUEST_19_EVENT_31_RANDOM_TEXT_5
; QST_QUEST_19_EVENT_31_RANDOM_TEXT_7
; QST_QUEST_19_EVENT_SEASON_AIR_TEXT
; QST_QUEST_19_EVENT_BUSH_TEXT


code:
MAP_NAME = "街角"
MAP_IMAGE = "街角.webp"
PATH_COLOR = "0x602c1c"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0x40545c"

repo_name: napo/erafl

license: unknown


code:
<DocumentElement>
  <GMAPDATA>
    <id>638476960071087196</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>並木道</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>0,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960136311041</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>賑わいの道</NODE_NAME>
    <POS_X>320</POS_X>
    <POS_Y>180</POS_Y>
    <PATH_LIST>1,3,6,4,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960179778017</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>少し静かな道</NODE_NAME>
    <POS_X>220</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960226457853</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>店の立ち並ぶ通り</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>2,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960246309688</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>怪しい店</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>4,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960276166151</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>公園</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>2,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960582149774</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>公園奥</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>6,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638476960652857511</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>町の入口</NODE_NAME>
    <POS_X>520</POS_X>
    <POS_Y>40</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_1:サンプルクエスト
;-------------------------------------------------
@QST_TITLE_QUEST_1
RESULTS '= "蠢く洞窟"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_1
RESULTS '= "単発,探索,"

;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_1
RESULTS:0 '= "森林で活動していたノルディアのアカデミーの植物採集隊が現地の洞窟で消息を絶った"
RESULTS:1 '= "依頼内容は彼らの救助である。成功すればアカデミーから報酬が支払われる"
RESULTS:2 '= ""
RESULTS:3 '= "しかし、十分に注意すべきだろう。先行して救助に向かった冒険者の一団も音信がない"
RESULTS:4 '= "何か未知の脅威があるようだ……"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_1
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★★★@/C@"
RESULTS:1 '= "報酬：15,000ディナリ"
RESULTS:2 '= "推奨パーティ人数:3人以上　　　推奨平均戦闘値：35以上"
RESULTS:5 '= @"@C:{カラー_選択不可}@「コモンウェルスはノルディアの学者さんらは穴があれば何処へでも頭を突っ込みたがる」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「……お得意様をあまり悪くは言えんが、大事な頭がもげちまったら元も子もなかろうに……」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_1(QUEST_ID)
#DIM QUEST_ID
;平原の探索の攻略が必要
SIF !QST_SV_QUEST_FLAG:3
    RETURN 0
SIF QST_SV_QUEST_FLAG:QUEST_ID != 2
    RETURN 1

;-------------------------------------------------
;マップデータ定義
;[ROOM_ID:]　…　配置するエリアの種類
;[ROOM_PATH:]　…　エリア間の接続情報(※)
;[EVENT_LIST:]　…　エリアに配置するイベントのID(※)
;(※)要素の末尾の','を忘れないように注意
;EmueraをDEBUGMODEで起動した場合、[領地の手入れ]画面から編集したエリア情報をテキストファイルにエクスポートできる
;出力txtをベースに編集すると少し楽かもしれない
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_1(MAP_ID)
#DIM MAP_ID
QST_MAPNAME '= "蠢く洞窟"
_DIC_HO_MAPDATA:MAP_ID:0  '= "[ROOM_ID:200][ROOM_PATH:5,]"
_DIC_HO_MAPDATA:MAP_ID:1  '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:2  '= "[ROOM_ID:202][ROOM_PATH:7,][EVENT_LIST:3,]"
_DIC_HO_MAPDATA:MAP_ID:3  '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:4  '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:5  '= "[ROOM_ID:202][ROOM_PATH:6,0,][EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:6  '= "[ROOM_ID:202][ROOM_PATH:7,5,][EVENT_LIST:1,]"
_DIC_HO_MAPDATA:MAP_ID:7  '= "[ROOM_ID:201][ROOM_PATH:2,8,12,6,]"
_DIC_HO_MAPDATA:MAP_ID:8  '= "[ROOM_ID:201][ROOM_PATH:7,][EVENT_LIST:4,]"
_DIC_HO_MAPDATA:MAP_ID:9  '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:10 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:11 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:12 '= "[ROOM_ID:202][ROOM_PATH:7,17,]"
_DIC_HO_MAPDATA:MAP_ID:13 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:14 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:15 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:16 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:17 '= "[ROOM_ID:202][ROOM_PATH:12,22,][EVENT_LIST:5,]"
_DIC_HO_MAPDATA:MAP_ID:18 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:19 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:20 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:21 '= "[ROOM_ID:0][ROOM_PATH:]"
_DIC_HO_MAPDATA:MAP_ID:22 '= "[ROOM_ID:201][ROOM_PATH:17,23,][EVENT_LIST:7,]"
_DIC_HO_MAPDATA:MAP_ID:23 '= "[ROOM_ID:202][ROOM_PATH:22,24,][EVENT_LIST:8,9,]"
_DIC_HO_MAPDATA:MAP_ID:24 '= "[ROOM_ID:201][ROOM_PATH:23,][EVENT_LIST:10,]"

;-------------------------------------------------
;クエストのタイプ
;マップ探索を伴う場合 "MAP"
;文章のみで完結する場合 "DIALOGUE"
;-------------------------------------------------
@QST_TYPE_QUEST_1
RESULTS '= "MAP"

;-------------------------------------------------
;QST_INTRO_QUEST_n はQUEST開始前に呼ばれる
;クエストの導入文などを記述
;"DIALOGUE"型のクエストは当該関数のCALLのみ行う
;-------------------------------------------------
@QST_INTRO_QUEST_1
PRINTFORMW %QST_LEADER_NAME()%の前にぽっかりと洞窟が口を開けている
PRINTFORMW 一見はなんの変哲もない洞窟だ
PRINTFORMW しかしこの場所でまずはアカデミーの植物採集隊が、次いで救助に向かった冒険者の一団が、相次いで行方不明になっているという
PRINTL
PRINTFORMW 中がどうなっているのか、どれほどの深さなのか、外からでは検討もつかない
PRINTFORMW %QST_LEADER_NAME()%がゆっくりと足を踏み入れると洞窟の奥から生ぬるい風が吹き込んだ
PRINTFORMW ひどく嫌な予感がする。十分に警戒して進むべきだろう……
PRINTW

;-------------------------------------------------
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

;-------------------------------------------------
;実行するイベントの実行内容を記述
;イベント用の関数はTRYCALLFORMで呼ばれるため
;@QST_QUEST_{QUEST_ID}_EVENT_{EVENT_ID}の書式で記述する
;-------------------------------------------------
@QST_QUEST_1_EVENT_0

PRINTFORMW %ANAME(QST_RANDCHARA())%は額を伝う汗を拭った
PRINTFORMW 洞窟の中は強い湿気を帯びており、異様なまでに生暖かい
PRINTFORMW 加えてかすかな生臭さが鼻をつく……

;-------------------------------------------------
;EV1
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_1
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
RESULTS = %EVENT_TRIGGER%

;-------------------------------------------------
;実行するイベントの実行内容を記述
;イベント用の関数はTRYCALLFORMで呼ばれるため
;@QST_QUEST_{QUEST_ID}_EVENT_{EVENT_ID}の書式で記述する
;-------------------------------------------------
@QST_QUEST_1_EVENT_1
#DIM DYNAMIC TARGET_ID
#DIM DYNAMIC MOB_ID
#DIM CONST 派生イベントID = 2

;派生先イベントが解決済みの場合イベント実行を省略
CALL QST_GET_SV_EVENT_FLAG(派生イベントID)
IF RESULT
    RETURN
ENDIF

TARGET_ID = QST_RANDCHARA()

PRINTFORMW 一行が隘路を奥へ進むと、隘路いっぱいに広がった水たまりに出くわした
PRINTFORMW 床を満たす液体は、暗くて十分には判別出来ないが、濁っており生臭い臭気を放っている
PRINTFORMW 洞窟を満たす異様な香りはこれのせいなのだろうか……？
PRINTL
PRINTFORMW いずれにせよ先に進むにはこの水たまりを横切るしかない
PRINTFORMW 深さは大人の膝下程あるようだ。中に何か潜んでいる可能性も考慮すべきだろう……
PRINTW
PRINTFORMW %ANAME(TARGET_ID)%が粘性を帯びた水たまりをかき分けて進んでいると何かが足先に触れた
PRINTFORMW 正体を探るべく、身を屈めてランタンを近づける……
PRINTW
CALL COLOR_PRINTW(@"―――― 何かが%ANAME(TARGET_ID)%の足に絡みつき、水たまりに引きずり込もうとしている！", カラー_注意)
CALL QST_ROLL_MULTI("戦闘,身体,", "ナイフで斬りつける(戦闘),全力で振りほどく(身体),","100,100,", , ,"なすがまま", TARGET_ID)
IF RESULT == 1
    CALL QST_QUEST_1_EVENT_2
ELSE
    CALL COLOR_PRINTW(@"%ANAME(TARGET_ID)%は触手に引きずり込まれ、引き倒されてしまった……", カラー_ピンク)
    ;調教イベントの初期設定
    CALL CLEAR_MPLY()
    CALL CLEAR_MTAR()
    FLAG:ウフフフラグ = 1
    FLAG:調教モード = 調教_慰安
    TFLAG:調教ターン数 = 5
    CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
    CFLAG:TARGET_ID:主導除外 = 1
    CFLAG:TARGET_ID:凌辱被害者フラグ = 1
    CFLAG:TARGET_ID:調教参加フラグ = 1

    ;モブ生成
    FOR LOCAL, 0, 3
        CALL QST_TRAIN_MOB_PRESET("ブラシ触手", "うねる茎")
        MOB_ID = RESULT
        CALL ADD_MPLY(MOB_ID)
        CALL SET_EQUIP(GETNUM(TRAINNAME, "触手召喚"), 1)
        CALL DEL_MPLY(MOB_ID)
    NEXT

    QUEST_1_TARGET_ID = TARGET_ID
    CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
ENDIF

;-------------------------------------------------
;EV2
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_2
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_1_EVENT_2
#DIM CONST EVENT_ID = 2
#DIM EVENT_FLAG
#DIM CC_RESULT
PRINTFORMW %ANAME(QUEST_1_TARGET_ID)%は触手をなんとか振りほどいた！
PRINTFORMW %QST_LEADER_NAME()%は武器を手に取り、なお襲いかからんとする触手の群れに対峙した
;戦闘イベント
CALL CC_INIT()
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
NEXT
CALL CC_ADDMOB("DEFENDER", "這い寄る触手", 2300, 2300, 30)
CALL CC_ADDMOB("DEFENDER", "滴り蠢く茎", 1800, 1800, 25)
CALL CC_ADDMOB("DEFENDER", "粘性触手", 1800, 1800, 25)
CALL CLOSE_COMBAT()
CC_RESULT = RESULT
CALL CC_APPLY_CHARA_HP()
;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
IF CC_RESULT == 1
    PRINTFORMW %QST_LEADER_NAME()%は襲いくる触手の群れを退けた
    CALL CC_ADD_COMBAT_EXP(2)
    CALL QST_SET_SV_EVENT_FLAG(EVENT_ID ,1)
ELSE
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%は触手の群れに打ち負かされてしまった……", カラー_注意)
    PRINTFORMW 抵抗する力を失った%QST_LEADER_NAME()%に無数の触手の群れが群がる……
    CALL QST_QUEST_1_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;EV3
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_3
#DIM CONST EVENT_ID = 3
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "発光する肉腫"

CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
SIF RESULT
    EVENT_SLOT_NAME '= "肉腫の残骸"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_1_EVENT_3
#DIM CONST EVENT_ID = 3
#DIM EVENT_FLAG
#DIM DYNAMIC CHARA_ID

CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%の前にある肉腫は仄かに光を帯びている
    PRINTFORMW 周囲に巻き付く触手を取り除けば、内部にあるものを取り出すことが出来そうだが……
    PRINTL
    CALL ASK_YN("回収を試みる (観察判定)","放っておく")
    SIF RESULT
        RETURN
    CALL QST_PARTY_ROLL("観察", 0, 1)
    CHARA_ID = RESULT:1
    IF RESULT == 1
        PRINTFORMW %ANAME(CHARA_ID)%は肉腫を覆う触手が食虫植物の感覚毛に近い器官を持つことに気がついた
        PRINTFORMW それに触れぬよう注意深く肉腫を切り開く……
        PRINTW
        PRINTFORMW 肉腫は高純度のマナ結晶を包んでいた。もしかしたらこれが彼らの栄養源なのかもしれない……
        CALL CHANGE_ITEM_STOCK(ITEM_マナ結晶, 6)
        CALL GET_ITEM_NAME(ITEM_マナ結晶)
        CALL COLOR_PRINTW(@"%RESULTS%を６つ手に入れた", カラー_注意)
        CALL QST_SET_EVENT_SLOT_NAME(EVENT_ID, "肉腫の残骸")
        CALL QST_SET_SV_EVENT_FLAG(EVENT_ID, 1)
    ELSEIF RESULT == 0
        CALL COLOR_PRINTW(@"―――― 肉腫を切り開こうと手をのばすなり、鞭のようにしなる触手が%ANAME(CHARA_ID)%を殴打した！", カラー_注意)
        CALL QST_CHANGE_HP(CHARA_ID, -200)
        CALL QST_ARMOR_BREAK(CHARA_ID, 被覆部位_乳首, 100)
    ENDIF
ELSE
    PRINTFORMW 切り開かれた肉腫の残骸が転がっている
ENDIF

;-------------------------------------------------
;EV4
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_4
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "湧き水"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_1_EVENT_4
#DIM CONST EVENT_ID = 4
#DIM EVENT_FLAG
#DIM TARGET_ID

CALL QST_GET_EVENT_FLAG(EVENT_ID)
EVENT_FLAG = RESULT
IF EVENT_FLAG == 2
    PRINTFORMW 水場は既に枯れている
    RETURN
ENDIF

PRINTFORMW 岩の隙間から水がしみ出している
PRINTFORMW 一見すると透明で危険はないように見えるが……
PRINTL
CALL ASK_MULTI_JUDGE("飲んでみる", 1, "よく調べる (観察判定)",EVENT_FLAG == 1 ? 0 # 1, "手を付けない", 1)
SELECTCASE RESULT
    CASE 0
        PRINTFORMW %QST_LEADER_NAME()%は水を手に取り口に含んだ
        PRINTFORMW 清純な水分が喉を潤す……
        ;パーティ全員の体力を回復
        FOR LOCAL, 0, QST_PARTYNUM
            TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
            CALL QST_CHANGE_HP(TARGET_ID, 500)
        NEXT
        WAIT
        CALL QST_SET_EVENT_SLOT_NAME(EVENT_ID, "枯れた水場")
        CALL QST_SET_EVENT_FLAG(EVENT_ID, 2)
    CASE 1
        CALL QST_PARTY_ROLL("観察", -200, 1)
        TARGET_ID = RESULT:1
        IF RESULT
            PRINTFORMW この水は非常に清潔なようだ！　飲めば消耗した体力を回復(+500)できるだろう
            CALL PRINT_ADD_EXP(TARGET_ID, "知識経験値", 4, 1)
            CALL TRAIN_AUTO_ABLUP(TARGET_ID)
            CALL QST_SET_SV_EVENT_FLAG(EVENT_ID, 2)
        ELSE
            PRINTFORMW %ANAME(TARGET_ID)%が確かめてみたもののよくわからなかった……
        ENDIF
        CALL QST_SET_EVENT_FLAG(EVENT_ID, 1)
    CASEELSE
        RETURN
ENDSELECT

;-------------------------------------------------
;EV5
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_5
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_1_EVENT_5
#DIM CONST EVENT_ID = 5
#DIM CONST 派生イベントID = 6
#DIM EVENT_FLAG
#DIM MOB_ID

CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
SIF RESULT
    RETURN

CALLF TAG_PRINT(@"@C:{カラー_ピンク}@ジュル……グチュッ……ジュルルッ……@/C@", 1)
WAIT
PRINTFORMW %QST_LEADER_NAME()%が洞窟を奥へ奥へと歩みをすすめると、奥から断続的な水音が聞こえてきた
PRINTFORMW 正体を確かめるべくさらに奥へと歩みを進める……
PRINTW
PRINTFORMW %QST_LEADER_NAME()%の眼前では凄惨な狂宴が繰り広げられていた

FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 12
FLAG:調教スキップ可 = 1

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
;モブ生成
FOR LOCAL, 0, 4
    CALL QST_TRAIN_MOB_PRESET(SPLIT_RAND("触手/触手/ブラシ触手",1), SPLIT_RAND("蠢く触手/粘性触手/触手群体",1))
    MOB_ID = RESULT
    CALL ADD_MPLY(MOB_ID)
    CALL SET_EQUIP(GETNUM(TRAINNAME, "触手召喚"), 1)
    CALL DEL_MPLY(MOB_ID)
NEXT
CALL QST_TRAIN_MOB_PRESET("触手寄生体", SPLIT_RAND("寄生体",1))
CALL QST_TRAIN_MOB_PRESET("触手寄生体", SPLIT_RAND("寄生体",1))

;調教対象を生成
CALL QST_QUEST_1_CREATE_TRAIN_VICTIMS

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)

CALL QST_SET_SV_EVENT_FLAG(EVENT_ID ,1)
RETURN

;-------------------------------------------------
;EV6
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_6
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_1_EVENT_6
PRINTFORMW %QST_LEADER_NAME()%は助けに入ろうと武器を握りしめる
PRINTFORMW ところが触手の群れは捕らえた女性をきつく絡め取り、洞窟のさらに深奥へと引き込んでしまった……

;-------------------------------------------------
;EV7
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_7
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_1_EVENT_7
#DIM CONST EVENT_ID = 7
#DIM CC_RESULT

CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
SIF RESULT
    RETURN

PRINTFORMW 隘路を抜けると少し開けた空間に出た
PRINTFORMW 先程の女性の姿は見えない……。代わりに二人の男が背を向けて立っている
PRINTL
PRINTFORMW 軽装の鎧を身にまとっており、その風貌からどうやら冒険者のように見える
PRINTFORMW 先立って救助に向かった者たちだろうか……？
PRINTL
PRINTFORMW %QST_LEADER_NAME()%は声をかけようとしたがどうも様子がおかしい……
PRINTFORMW 「アッ……アア……、アァ……」
PRINTFORMW 男たちはうわ言のように何かを呟いている
PRINTL
PRINTFORMW %ANAME(QST_RANDCHARA())%は男の鎧の隙間から触手が蠢くのを目にしてしまった。彼らは無数の触手に寄生されている！
PRINTFORMW 男たちはゆっくりと振り返り、剣先を%QST_LEADER_NAME()%に向けた
PRINTL
PRINTFORMW ―― 既に彼らは正気ではない。どうやら戦うしかないようだ！
PRINTW

;戦闘イベント
CALL CC_INIT()
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
NEXT
CALL CC_ADDMOB("DEFENDER", "寄生体", 2500, 2500, 35)
CALL CC_ADDMOB("DEFENDER", "寄生体", 2500, 2500, 35)
CALL CLOSE_COMBAT()
CC_RESULT = RESULT
CALL CC_APPLY_CHARA_HP()
IF CC_RESULT == 1
    PRINTFORMW %QST_LEADER_NAME()%は寄生体を打ち倒した
    CALL CC_ADD_COMBAT_EXP(3)
    WAIT
    PRINTL
    PRINTFORMW さらに冒険者の亡骸から使えそうな装備を回収した
    CALL QST_GET_WEAPON(1, "冒険者の鉄の短剣", WPN_材質_鉄)
    CALL COLOR_PRINTW(@"―― 武器はクエスト中でも[ITEM] -> [武器]から変更できる。状況に応じて持ち替えるとよいだろう", カラー_注意)
    CALL QST_SET_SV_EVENT_FLAG(EVENT_ID ,1)
ELSE
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%は寄生体に打ち負かされてしまった……", カラー_注意)
    PRINTFORMW 抵抗する力を失った%QST_LEADER_NAME()%に無数の触手の群れが群がる……
    CALL QST_QUEST_1_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;EV8
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_8
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_1_EVENT_8
#DIM CONST EVENT_ID = 8
#DIM CONST 派生イベントID = 9
CALL QST_GET_SV_EVENT_FLAG(派生イベントID)
SIF RESULT == 2
    RETURN

PRINTFORMW %QST_LEADER_NAME()%の前に床に打ち捨てられた一糸まとわぬ女性が横たわっている
PRINTFORMW 先程の寄生体の件もある。軽率に近寄るのは避けるべきかもしれないが……

;-------------------------------------------------
;EV9
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_9
#DIM CONST EVENT_ID = 9
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "全裸で横たわる女性"
CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
SIF RESULT == 2
    EVENT_SLOT_NAME '= "女魔術師"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_1_EVENT_9
#DIM CONST EVENT_ID = 9
#DIM EVENT_FLAG
#DIM TARGET_ID

CALL QST_GET_SV_EVENT_FLAG(EVENT_ID)
EVENT_FLAG = RESULT
IF EVENT_FLAG == 2
    PRINTFORMW 「どうかお気をつけて……」
    RETURN
ENDIF

PRINTFORMW さて、どうしたものだろうか……？
PRINTL
CALL ASK_MULTI_JUDGE("救助を試みる", 1, "よく観察する (観察判定)",EVENT_FLAG == 1 ? 0 # 1, "放っておく", 1)
SELECTCASE RESULT
    CASE 0
        PRINTFORMW 幸いなことに女性は触手の寄生を受けておらず、息もあるようだ
        PRINTFORMW 気付け薬を少し飲ませてやると、意識を取り戻した
        PRINTL
        PRINTFORMW 「助かった……のでしょうか……？」
        PRINTL
        PRINTFORMW %QST_LEADER_NAME()%は女性が落ち着くのを待ってから情報収集を試みる
        PRINTFORMW 「生存者は……おそらく私だけでしょう。皆あれに喰われてしまいました……。私が最後の生き餌だったのです」
        PRINTFORMW どうやら女性は学術調査隊に同行していた護衛の魔術師のようだ
        PRINTL
        PRINTFORMW 「あれらの触手には母体が存在します。母体を葬れば末端の触手もやがて枯死するでしょう」
        PRINTFORMW 「どうか隊の……皆の仇をとっていただけないでしょうか……？」
        PRINTL
        PRINTFORMW そのために来たのだと%ANAME(QST_RANDCHARA())%が告げると女性は安堵の表情を浮かべた
        PRINTW
        PRINTFORMW 「ありがとうございます……。母体は常に無数の小群体を護衛として引き連れています」
        PRINTFORMW 「小群体は攻撃的ですが、脆弱です。あれを優先して排除すれば優位に立てるはずです」
        PRINTL
        CALL COLOR_PRINTW(@"―― 脅威選定に(低耐久)を持つ武器は体力の低い敵を優先的に攻撃する。所持しているならば装備することで優位に戦闘を進められるかもしれない", カラー_注意)
        PRINTL
        PRINTFORMW 女魔術師が呪文を唱えると%QST_LEADER_NAME()%は身体が軽くなるのを感じた
        PRINTL
        FOR LOCAL, 0, QST_PARTYNUM
            TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
            CALL QST_CHANGE_HP(TARGET_ID, 1500)
        NEXT
        PRINTL
        WAIT
        PRINTFORMW 「どうかご武運を……」
        CALL QST_SET_EVENT_SLOT_NAME(EVENT_ID, "女魔術師")
        CALL QST_SET_SV_EVENT_FLAG(EVENT_ID, 2)
    CASE 1
        CALL QST_PARTY_ROLL("観察", 0, 1)
        IF RESULT
            PRINTFORMW どうやら寄生はされておらず、息もあるようだ。今なら助けられるだろう
        ELSE
            PRINTFORMW %ANAME(RESULT:1)%には判断がつかなかった……。
        ENDIF
        CALL QST_SET_SV_EVENT_FLAG(EVENT_ID, 1)
    CASEELSE
        RETURN
ENDSELECT

;-------------------------------------------------
;EV10
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_10
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_1_EVENT_10
#DIM CONST EVENT_ID = 10
#DIM CC_RESULT

PRINTFORMW 洞窟の深奥に歩みを進めた%QST_LEADER_NAME()%を猛烈な熱気と臭気が襲う
PRINTFORMW 空間の中央に鎮座しているのは、禍々しい肉のうねりだ
PRINTFORMW 無数の触手が蠢き、のたうち、一つの巨大な群体を構成している
PRINTL
PRINTFORMW さらにその周囲をせわしなく動く小型の触手の群れが這い回っている
PRINTFORMW 母体を護衛する役割を担っているようだ
PRINTL
PRINTFORMW %ANAME(QST_RANDCHARA())%は眼前のおぞましい光景に顔をしかめ、固く武器を握りしめる
PRINTFORMW これら全てを退治するのはいささか骨が折れそうだ……

;戦闘イベント
CALL CC_INIT()
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
NEXT
CALL CC_ADDMOB("DEFENDER", "テンタクル・レギオン", 3500, 3500, 40)
CALL CC_ADDMOB("DEFENDER", "繊毛の尖兵", 300, 300, 30)
CALL CC_ADDMOB("DEFENDER", "破甲の茎", 300, 300, 30)
CALL CC_ADDMOB("DEFENDER", "浸透する細毛", 300, 300, 30)
CALL CC_ADDMOB("DEFENDER", "刺突毛", 300, 300, 30)
CALL CC_ADDMOB("DEFENDER", "乱舞する鞭毛", 300, 300, 30)
CALL CLOSE_COMBAT("[BGM:Boss_0.ogg]")
CC_RESULT = RESULT
CALL CC_APPLY_CHARA_HP()
IF CC_RESULT == 1
    PRINTFORMW %QST_LEADER_NAME()%は死闘の末に触手の群れを打ち倒した……！
    CALL CC_ADD_COMBAT_EXP(5)
    CALL QST_SET_SV_EVENT_FLAG(EVENT_ID ,1)
    PRINTW
    PRINTFORMW %QST_LEADER_NAME()%の一撃が致命傷となり、群体を維持できなくなったテンタクル・レギオンは崩れ落ちた
    PRINTFORMW 周囲に展開していた無数の触手も母体からの栄養の供給を失い、程なくして動かなくなった
    PRINTW
    CALL FL_PLAYSE("Fanfare_0.ogg")
    CALL COLOR_PRINTW(@"―― %QST_LEADER_NAME()%は依頼を完遂した！", カラー_注意)
    CALL QST_QUEST_1_ENDING
ELSE
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%は触手の群れに打ち負かされてしまった……", カラー_注意)
    PRINTFORMW 抵抗する力を失った%QST_LEADER_NAME()%に無数の触手の群れが群がる……
    CALL QST_QUEST_1_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;EV15
;敗北イベント後の共通エンディング（クエスト失敗）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_1_EVENT_15
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_1_EVENT_15
#DIM CONST EVENT_ID = 15
#DIM EVENT_FLAG
#DIM TARGET_ID
TARGET_ID = QST_RANDCHARA()
PRINTW
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTW
PRINTFORMW 事態を重く見たギルドが派遣した熟練冒険者たちによって%QST_LEADER_NAME()%は奇跡的に救出された……
PRINTFORMW 「しかしまぁ……生き残っちまった方がよほど不運ですぜ、ありゃぁ」
PRINTFORMW 毛布にくるまり焦点のあわない虚ろな目で荷馬車に揺られる%ANAME(TARGET_ID)%に冒険者の一人が憐れみの目を向ける
PRINTFORMW 「よしたまえよ。……まぁ何れにせよ命は残った。再起できるかは……%PRONOUN(TARGET_ID)%次第だろう」
PRINTW
PRINTFORMW %QST_LEADER_NAME()%はなんとか生還することができた……
PRINTFORMW クエストは失敗に終わったが、再起の機会はあるはずだ
PRINTL
QST_SV_QUEST_FLAG:QST_有効クエストID = 1
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
@QST_QUEST_1_ENDING
PRINTW
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTW
PRINTFORMW %QST_LEADER_NAME()%たちは無事依頼を完遂し帰路についた
PRINTFORMW あの恐るべき触手の怪物がどのようして生じたのか、あるいはどこからやってきたのか、以前謎は残る
PRINTFORMW しかし一先ずは、これ以上哀れな被害者を増やすことは防げたはずだ
PRINTW
PRINTFORMW 後日、ノルディアのアカデミーからギルドを介して%QST_LEADER_NAME()%に報酬が支払われた
CALL COLOR_PRINTW("15,000ディナリを得た", カラー_注意)
MONEY += 15000
QST_SV_QUEST_FLAG:QST_有効クエストID = 2
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------

@QST_QUEST_1_DEFEAT_EVENT
#DIM CONST 派生イベントID = 15
#DIM DYNAMIC MOB_ID

;調教イベントの初期設定
CALL CLEAR_MPLY()
CALL CLEAR_MTAR()
FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 20
CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):主導除外 = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):凌辱被害者フラグ = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):調教参加フラグ = 1
NEXT

;触手系モブ生成
FOR LOCAL, 0, 4
    CALL QST_TRAIN_MOB_PRESET(SPLIT_RAND("触手/触手/ブラシ触手",1), SPLIT_RAND("蠢く触手/粘性触手/触手群体",1))
    MOB_ID = RESULT
    CALL ADD_MPLY(MOB_ID)
    CALL SET_EQUIP(GETNUM(TRAINNAME, "触手召喚"), 1)
    CALL DEL_MPLY(MOB_ID)
NEXT
CALL QST_TRAIN_MOB_PRESET("触手寄生体", SPLIT_RAND("寄生体",1))

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
RETURN
;-------------------------------------------------
;慰安モブの定義
;-------------------------------------------------
@QST_QUEST_1_CREATE_TRAIN_VICTIMS(主導除外フラグ = 1)
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %SPLIT_RAND("冒険者/村娘/魔術師/学者",1)%

CALL CREATE_RANDOM_CHARA(0, -1, 1)
FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
CALLNAME:対象 = %名前%
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = 1
ABL:対象:性知識 =  3
ABL:対象:Ｃ感 = 28
ABL:対象:奉仕 = 1
ABL:対象:性技 = RAND(10, 14)
ABL:対象:欲望 = RAND(3, 7)
ABL:対象:性交 = RAND(13, 15)
ABL:対象:マゾ = RAND(0, 3)
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = RAND(0, 300)
TALENT:対象:貞操無頓着 = 1
TALENT:対象:貞操観念 = 0
TALENT:対象:崩壊 = 1
TALENT:対象:母乳体質 = 1
CFLAG:対象:調教参加フラグ  = 主導除外フラグ
CFLAG:対象:主導除外  = 1
CFLAG:対象:慰安モブ = 1